

Randomization of Subjects

Subjects are assigned to study groups by a random mechanism not controlled by the patient or the investigator. Randomization increases the likelihood treatment groups are comparable with respect to distribution of measurable and measurable characteristics.

Randomization Schedule

Prior to study start, a randomization schedule is generated by a statistician. Ex. a 2 treatment trial with 1:1 randomization.

Treatment allocation schemes should:

- Be unpredictable
 - Neither the participant nor the investigator know in advance which treatment will be assigned (reduced observation bias)
- Promote balance
 - Groups must be alike in all important aspects and only differ in the intervention each group receives
- Be simple
 - Easy for investigator/staff to implement

Generally, the randomization ratio should be equally allocated (1:1 randomization). Unequal randomization arises in some situations where we want more in one group than another due to costs or other factors.

Methods of Randomization

- Unrestricted randomization
- Block randomization
 - Randomized permuted blocks
- Stratification
- Adaptive randomization
 - Biased coin, etc
 - Minimization
- Cluster randomization

Unrestricted Randomization

- Equivalent to tossing a fair coin for each subject that enters the trial
- Assign each treatment randomly and independently of previous treatment

- Also called simple/complete randomization

SAS Code for generating unrestricted random data:

```
*Method 1;
*RANUNI: generates random numbers between 0 and 1 which have a uniform distribution;
data one;
seed=123;
do id=1 to 8;
r=ranuni(seed);
if r<0.5 then group='A';
if r>=0.5 then group='B';
output;
end;
run;

* Method 2;
data one;
seed=123;
do id=1 to 8;
r=ranuni(seed);
output;
end;
run;
proc sort data=one;
by r;
run;
data two;
set one;
if _n_<=4 then group='A';
if _n_>4 then group='B';
run;
proc sort data=two;
by id; * Sort again by id;
run;

* Method 3;
data three;
seed=123;
do id=1 to 8;
```

```

r=ranuni(seed);
output;
end;
run;

proc rank data=three groups=3 out=three;
var r;
ranks group;
run;

data three;
set three;
if group=0 then trt='A';
if group=1 then trt='B';
if group=2 then trt='C';
run

```

Note that the seed is a number that determines the starting value. The same positive seed will always generate the same result. If we want a different number each run then set the seed ≤ 0 .

The issue with unrestricted randomization is that it can produce imbalance. Ex. Based on the binomial distribution, a study with two groups and 8 patients will produce a 6:2 assignment 28% of the time.

Block Randomization

Also called permuted block, and is the most common form of randomization in clinical trials. Sequence of blocks that contain the assignment in the desired ratio.

Ex. With $N = 8$ patients and 2 treatments, A and B, with 1:1 randomization and block size 4:

1. Take the letters AABB, randomly permute them
2. Assign treatment for the next set of 4 patients, randomly permute AABB again

Block randomization ensures that of the first 4 patients, 2 receive each treatment.

Choosing Block Size

The block size must be a multiple of the number in the allocation ratio. Block size is not always mentioned in the study's protocol, as no one needs to know except the statistician who generates the schedule.

A block size of 2 is usually not recommended. It creates a "guessable" pattern of AB, BA, AB...

A large block size is also not recommended (especially for small samples). May lead to incomplete blocks at the end of the study. Ex. $N = 12$ with block size 8 would leave 4 incomplete blocks.

Block randomization SAS with 12 patients, 2 treatments and block size 4:

```
proc plan seed=1133637;
factors blocks=3 ordered trt=4 random/noprint;
output out=randschd trt cvals=('A' 'A' 'B' 'B');
run; quit;

ods rtf file='rand.rtf';
options nodate nonumber;
* Add patient ID ;
data randschd;
set randschd;
patient+1;
run;
* Final randomization schedule;
proc print data=randschd (obs=12)
label noobs;
var patient trt;
label patient='Patient ID'
trt='Treatment';
title1 'Pharmaceutical
Company, Inc. ';
title2 'Protocol XXXXXX
Randomization Schedule';
run;
ods rtf close;
```

Random Block-Size Randomization

In 'open label' studies there is no concern the investigator might deduce the block size over time.

In this method we choose 2 block sizes and randomly assign a size to each block. Then within each block, randomize patients to treatment using permuted block method. This makes it very difficult to guess the next assignment.

Ex. ABBABA - BABA - AABBA - BABABA

```
* Random block with 2 treatments, 20 participants and block sizes of 2 and 4 with 1:1 randomization;
* Step 1: generate blocks of random size;
data blocks;
do block=1 to 10;
```

```

half_size=rantbl(2019,0.5,0.5);
size=half_size*2;
do j=1 to half_size;
  do treatment='A','B';
    random=ranuni(2019);
    output;
  end;
end;
end;
run;

* Step 2: permute the blocks;
proc sort data=blocks;
by block random;
run;

```

Stratified Randomization

Randomization is performed within strata by one or more subject characteristics. Use a permuted blocks design within each stratum. The goal is to ensure balance within each stratum.

In multi-center studies typically one stratifies the randomization by center.

Too many small strata can defeat the balancing effects of blocking, because you could get imbalances or long runs when you look at the sample as a whole.

- Use a limited number of strata in your randomization
- Consider using regression adjustments later if there are imbalances in baseline factors

The SAS code for 24 participants, 2 treatments, 1:1 randomization and block size 4.

```

proc plan seed=8894670;
factors site=2 ordered
blocks=3 ordered
trt=4 random/noprint;
output out=randschd2
trt cvals=('A' 'A' 'B' 'B');
run; quit;

* Create unique study ID ;
data randschd2;
set randschd2;

```

```

by site;
if first.site then patient=0;
patient+1;
if site=1 then
patid=patient+100;
if site=2 then
patid=patient+200;
run;

*Same as above, but with 2 weight stratum within each of the two sites;
proc plan seed=8894670;
factors site=2 ordered wgt=2 ordered blocks=3 ordered trt=4
random/noprint;
output out=randschd3 trt cvals=('A' 'A' 'B' 'B');
run; quit;

*2:1 Randomization with 18 participants, 2 treatments, block size 6 and 2 sites;
proc plan seed=8894670;
factors site=2 ordered wgt=2 ordered blocks=3 ordered
trt=6 random/noprint;
output out=randschd4 trt cvals=('A' 'A' 'A' 'A' 'B' 'B');
run; quit;

```

Adaptive Randomization

Array of methods to determine treatment assignment aimed at making trials more efficient. Allows to adapt the study design using data accumulated from early stages of the trial.

Two types:

- Based on the previous treatment assignments (e.g. Efron biased coin, minimization)
- Based on previous outcomes

Minimization is an alternative to stratification when the number of strata is large. The procedure is as follows:

1. k subjects are randomly assigned to treatment groups A or B
2. For each subsequent subject, compute an **imbalance score** under the two hypothetical treatment assignments A and B
3. Assign subject to the treatment allocation with lower score (deterministic approach) or give higher assignment probability to treatment with lower score (probabilistic approach).

The general idea is that as the data accumulates we will tend to assign subjects to the treatment that proves more effective.

'Play the Winner' Randomization Scheme:

1. If treatment A is successful based on accumulated data, a new available subject will be given a higher probability of being assigned to A
2. If treatment A is not successful then assign to B with higher probability
3. Proceed over time until we have enough evidence to make a decision

Pros and Cons:

- + Typically requires fewer subjects (efficiency)
- Special statistical methods
- Unlike permuted blocks, randomization lists cannot be made available before the study and require recalculation of treatment assignment probability with each new subject

Cluster Randomization

In studies of health service implementation or treatment delivery it is often preferable to randomize centers, physicians or communities rather than individual subjects.

Also used when there is the potential for contamination; aspects of intervention adopted by participants in the control group.

Blinding/Masking

Biases may be intentionally or unintentionally introduced into the conduct and analysis of a trial. Personnel and participants may have inherent biases regarding the treatment.

Types of Bias:

- Selection bias: Occurs when the investigator consciously or otherwise uses knowledge of the upcoming treatment assignment to help decide who to enroll
- Observer bias: Occurs when the knowledge of treatment allocation affects the evaluation of the response to treatment
 - Ex. A physician knows a patient receives treatment and subconsciously may probe more when evaluating
- Perception bias
 - Ex. A patient receives a control subconsciously may be less enthusiastic and under-report
 - an efficacious effect.

Blinding is the process of 'hiding' the treatment that a patient receives. When it is feasible, blinding eliminates potential biases.

Types of Blinding:

- Double-blind trial (most common) - Neither the study physician treating/evaluating nor the subject know the treatment patient is receiving
- Single-blind - Treating physician knows, but patient does not know
- Open-label - both subject and investigator know
- Triple-Blind - not an official phrase, but this includes the Sponsor or Statistician in the blinding

In single-blind and open-label we can't blind the treating physician but maybe another physician unaware of treatment allocation evaluate the subject for the efficacy/safety.

It is usually acceptable for the pharmacist or manufacture to be unblinded.

Revision #6

Created 27 January 2023 15:04:02 by Elkip

Updated 27 January 2023 17:30:40 by Elkip